
1
Introduction

In ENDC measurement gap configuration for EUTRA needs to take into account a potential measurement gap timing advance (mgta) as given in the extension of MeasConfig according to TS 36.331.
( MeasGapConfig_Type needs to be enhanced to configure the system simulator with mgta.

2
Proposal
As done for other cases (e.g. PRACH_Config_Type) a new “R8andLater” branch shall be added to MeasGapConfig_Type which for now can be used for ENDC. After a grace period (to be defined) the “R8” branch shall be removed and the “R8andLater” branch shall be used instead in all test suites.
Enhancement of MeasGapConfig_Type
	  type union MeasGapConfig_Type {

    MeasGapConfig R8,                                           /* !!!! to be removed after grace period !!!! */
    MeasGapConfig_R8andLater_Type R8andLater
  };

	  type record MeasGapConfig_R8andLater_Type {

    MeasGapConfig       measGapConfig,

    boolean             mgta  optional                          /* measurement gap timing advance for ENDC according to TS 36.331:

                                                                   omit means false at initial configuration and "keep as it is" otherwise */

  };


Change of cs_MeasGapCtrl_Config
	  template (value) MeasGapCtrl_Type cs_MeasGapCtrl_Config(template (value) MeasGapConfig p_MeasGapConfig ,

                                                          boolean p_Mgta := omit) :=

  { 
    Config := fl_MeasGapCtrl_Config(p_MeasGapConfig, p_Mgta)



  };

	  function fl_MeasGapCtrl_Config(template (value) MeasGapConfig p_MeasGapConfig,

                                 template (omit) boolean p_Mgta) return template (value) MeasGapConfig_Type 

  { /* !!!! temporary implementation !!!! */

    var template (value) MeasGapConfig_Type v_MeasGapConfig;

    if (f_GetTestcaseAttrib_ENDC(testcasename())) {

      v_MeasGapConfig := {

        R8andLater := {

          measGapConfig := p_MeasGapConfig,

          mgta := p_Mgta

        }

      };

    } else {

      v_MeasGapConfig := {

        R8 := p_MeasGapConfig

      };

    }

    return v_MeasGapConfig;

  }


Change of f_EUTRA_SkipMeasGap
	  function f_EUTRA_SkipMeasGap(EUTRA_CellId_Type p_CellId,

                               SubFrameTiming_Type p_SubFrameTiming) runs on EUTRA_PTC return SubFrameTiming_Type

  { /* calculation of measurment gap acc to 36.331 cl. 5.5.2.9 depending on configuration as defined in 36.133 Table 8.1.2.1-1 */

    var MeasGapCtrl_Type v_MeasGapCtrl := f_EUTRA_CellInfo_GetMeasGapCtrl(p_CellId);

    var SubFrameTiming_Type v_SubFrameTiming := p_SubFrameTiming;       // => per default the timing is not changed

    var MeasGapConfig.setup.gapOffset v_MeasGapOffset;

    var UInt_Type v_GapSfnPeriod;

    var UInt_Type v_GapOffset;

    var UInt_Type v_SfnBase;

    var UInt_Type v_GapStart;

    var UInt_Type v_GapEnd;

    var UInt_Type v_SFN;

    var UInt_Type v_Subframes;

    if (ischosen(v_MeasGapCtrl.Config.R8andLater.measGapConfig.setup)) {

      v_MeasGapOffset := v_MeasGapCtrl.Config.R8andLater.measGapConfig.setup.gapOffset;

      if (match(true, v_MeasGapCtrl.Config.R8andLater.mgta)) {

        FatalError(__FILE__, __LINE__, "f_EUTRA_SkipMeasGap: Measurement gap timing advance not supported for scheduled transmissions");

      }

    } else if (ischosen(v_MeasGapCtrl.Config.R8.setup)) {

      v_MeasGapOffset := v_MeasGapCtrl.Config.R8.setup.gapOffset;

    }
    if (isbound(v_MeasGapOffset)) {


      if (ischosen(v_MeasGapOffset.gp0)) {

        v_GapSfnPeriod := 4;

        v_GapOffset := v_MeasGapOffset.gp0;

      }

      else if (ischosen(v_MeasGapOffset.gp1)) {

        v_GapSfnPeriod := 8;

        v_GapOffset := v_MeasGapOffset.gp1;

      }

      else {

        FatalError(__FILE__, __LINE__, "f_EUTRA_SkipMeasGap: unknown MeasGap configuration");

      }

      v_SFN := p_SubFrameTiming.SFN.Number;

      v_Subframes := (v_SFN * 10) + p_SubFrameTiming.Subframe.Number;

      v_SfnBase := (v_SFN / v_GapSfnPeriod) * v_GapSfnPeriod;   // SFN rounded down to the next multiple of v_GapSfnPeriod

      v_GapStart := (v_SfnBase * 10) + v_GapOffset;             // start of the next gap in subframes

      v_GapEnd := v_GapStart + 6;                               // end of the next gap in subframes (acc. 36.133 Table 8.1.2.1-1 MeasurementGapLength is 6ms for both configurations)

      if ((v_GapStart <= v_Subframes) and (v_Subframes < v_GapEnd)) {

        v_SFN := (v_SFN + 1);  // @sic R5w160204r1 sic@

        v_SubFrameTiming.SFN.Number := v_SFN mod 1024;

        v_SubFrameTiming.HSFN.Number := (v_SubFrameTiming.HSFN.Number + (v_SFN / 1024)) mod 1024; // @sic R5w160204r1 sic@

      }

    }

    return v_SubFrameTiming;

  }


